Az integrálszámítás alaptétele

Ha az f függvénynek van elemi függvényként megadható primitív függvénye, akkor a határozott integrál ennek segítségével könnyen kiszámítható. A kiszámítás alapja a Newton-Leibniz tétel. A tételt a Lagrange-tétel felhasználásával bizonyítjuk. Ezért elõzõleg átismételjük az utóbbi tételt.

Lagrange-tétele

Tétel. (Lagrange tétele)

Ha $f(x)$ az $[a, b]$ zárt intervallumon folytonos és
az (a,b) nyílt intervallumon differenciálható,
akkor az (a,b) intervallumban van olyan ξ amelyre

$$(\mathbf{D}(f))(\xi)=\frac{f(b)-f(a)}{b-a}.$$

Másképp jelölve

$$\left.\left(\frac{d}{dx} f(x)\right)\right|_{x=\xi}=\frac{f(b)-f(a)}{b-a}$$

A tétel szerint a feltételeket teljesítõ f függvény esetén van az (a,b) intervallumban olyan ξ hely, ahol a függvény változási sebessége egyenlõ az $[a, b]$ intervallumbeli átlagos változási sebességgel. Geometriailag ez azt jelenti, hogy az adott pontbeli érintõ párhuzamos az intervallum végpontjaihoz tartozó szelõvel.
Az alábbi Langrange eljárás a Langrange tételt állítását szemlélteti. A feltételek teljesülése esetén ábrázolja a szelõt és a vele párhuzamos érintõt, egyébként pedig hibaüzenetet küld.

```
> restart:
> Lagrange:=proc(f,a,b)
  local abra1,abra2, vonal1, vonal2, xi, szelo:
  if not(iscont(f(x),x=a..b,'closed')) then
   ERROR(`Nem folytonos az [a,b] zárt intervallumon`):
  fi:
  if not(iscont(D(f)(x),x=a..b)) then
   ERROR(`Nem deriválható az (a,b) nyílt intervallumon`);
  fi:
  xi:=fsolve(D(f)(x)=(f(b)-f(a))/(b-a),x=a..b);
  abra1:=plot(f(x),x=a-0.1..b+0.1,color=blue):
  abra2:=plot(D(f)(xi)*(x-xi)+f(xi),x=xi-2..xi+2,color=red,
  thickness=2):
  vonal1:=plottools[line]([a,0],[a,f(a)],color=green,linestyle=3):
  vonal2:=plottools[line]([b,0],[b,f(b)],color=green,linestyle=3):
  szelo:=plot((f(b)-f(a))/(b-a)*(x-a)+f(a),x=a..b,color=magenta):
  plots[display]([vonal1,vonal2,szelo,abra1,abra2])
  end:
```

Hívjuk meg a Lagrenge eljárást a sin függvénnyel, a vizsgált intervallum legyen [-4,3] . Megjegyezzük, hogy a $f(x)=\sin(x)$ eljesíti a Lagrange tétel feltételeit, hiszen a színusz függvény a valós számos halmazán deriválható és így ott folytonos is.
Javasoljuk kipróbálni más intervallumokon a Langrange eljárást. Ehhez csak az alábbi utasításban szereplõ intervallum végpontokat kell megvaltoztatnunk és újra végrehajtani az utasítást (értsd. leütni az Enter billentyût).

```
> Lagrange(sin,-4,3);
```



Az eljárás szakaszonként adott (piecewise) függvényekre is alkalmazható:

```
> g:=x->piecewise(x<=0,-x*sin(x),x*sin(x));
```

$$g := x \to piecewise(x \leq 0,\ -x\sin(x),\ x\sin(x)) \qquad (1.1)$$

```
> Lagrange(g,-2,3);
```



Tekintsünk egy példát arra az esetre, amikor a feltételek valamelyike nem teljesül:

Példa

Vizsgáljuk meg, hogy teljesülnek-e a Lagrange-tétel feltételei az

$$f(x) = \begin{cases} -x^2 + 8x - 14 & x \le 5 \\ -\frac{1x}{2} + \frac{7}{2} & 5 < x \end{cases}$$

függvényre!

Megoldás

Vegyük föl a szakaszonként adott függvényt!

```
> f:=x->piecewise(x<=5,-x^2+8*x-14,x>5,x^2-7*x+11);
```

$$f := x \rightarrow piecewise\left(x \le 5,\ -x^2 + 8x - 14,\ 5 < x,\ x^2 - 7x + 11\right) \quad (1.2)$$

```
> 'f(x)'=f(x);
```

$$f(x) = \begin{cases} -x^2 + 8x - 14 & x \le 5 \\ x^2 - 7x + 11 & 5 < x \end{cases} \quad (1.3)$$

Ábrázoljuk a függvényt és deriváltját közös koordinátarendszerben:

```
> plot([f(x),D(f)(x)],x=4..6,discont=true,color=[blue,red],
  thickness=2,legend=["f(x)","D(f)(x)"]);
```



A függvény folytonos az $[5, 7]$ intervallumon, az $x = 5$ helyen azonban nem differenciálható.

```
> iscont(f(x),x=4..6);
```

$$true \tag{1.4}$$

```
> iscont(D(f)(x),x=4..6);
```

$$false \tag{1.5}$$

```
> Lagrange(f,4,6);
Error, (in Lagrange) Nem deriv&#225;lhat&#243; az (a,b) ny&#237;lt intervallumon
```

A következõ utasítássorozat különbözõ intervallumok egymásutánján **animáció**val szemlélteti a Langrange tétel állítását.

```
> f:=x->sin(x)+cos(2*x);
```

$$f := x \rightarrow \sin(x) + \cos(2\,x) \tag{1.6}$$

```
> p:=seq(Lagrange(f,-4,4.6-k*0.2),k=0..12):
```

```
> plots[display]([p],insequence=true);
```




```
>
```

Newton-Leibniz tétel

Tétel

Ha $f(x)$ az $[a, b]$ intervallumon integrálható (létezik határozott integrálja) és ezen az intervallumon primitív függvénye is van (legyen ezek egyike F), akkor

$$\int_a^b f(x)\,\mathrm{d}x = F(b) - F(a)$$

Bizonyítás

Tekintsük az $[a, b]$ intervallum tetszõleges beosztását. Tekintve, hogy a F függvény a f primitív függvénye az $[a, b]$ intervallumon, F deriválható az $[a, b]$ intervallumon, és annak bármely részintervallumán, s $\frac{d}{dx}F(x)=f(x)$ $(F'(x)=f(x))$. Ha a függvény deriválható, akkor folytonos. Ezért az F függvényre az $[a, b]$ bármely $\left[x_{i-1}, x_i\right]$ részintervallumán érvényes Lagrange tétele, tehát az $\left[x_{i-1}, x_i\right]$ intervallumnak van olyan belsõ ξ_i pontja, amelyre:

$$\frac{F\left(x_i\right)-f\left(x_{i-1}\right)}{x_i-x_{i-1}}=\left(\frac{d}{dx}F(x)\right)\Bigg|_{x=\xi_i}=f\left(\xi_i\right),$$

ahol $i=1, 2, ..., i, ..., n$.

Vegyük a Lagrange-tétel által meghatározott közelítõ összegeket, vagyis minden egyes részintervallumban a ξ_i kiválasztott változóérték legyen a Lagrange tételben szereplõ ξ_i :

$$\sigma(n)=\sum_{i=1}^{n}f\left(\xi_i\right)\left(x_i-x_{i-1}\right)=\sum_{i=1}^{n}\frac{F\left(x_i\right)-F\left(x_{i-1}\right)}{x_i-x_{i-1}}(x_i-x_{i-1})=\sum_{i=1}^{n}\left(F\left(x_i\right)-F\left(x_{i-1}\right)\right)=$$

$$=(F\left(x_1\right)-F\left(x_0\right))+(F\left(x_2\right)-F\left(x_1\right))+...+(F\left(x_{n-1}\right)-F\left(x_{n-2}\right))+(F\left(x_n\right)-F\left(x_{n-1}\right))=(F\left(x_n\right)-F\left(x_0\right))=$$

$$=F(b)-F(a)$$

Tehát kaptuk, hogy

$$\sigma(n)=\sum_{i=1}^{n}f\left(\xi_i\right)\left(x_i-x_{i-1}\right)=F(b)-F(a)$$

Ez azt jelenti, hogy az imént, Lagrange tétele alapján adódott speciális integrálközelítõ összegek azonosan egyenlõk az $F(b)-F(a)$ értékkel, tehát nyílván a határértékük is egyenlõ $F(b)-F(a)$-val. De mi a helyzet más közelítõ összegek választása esetén? Tudjuk, a feltétel szerint az f függvény integrálható az $[a, b]$ intervallumon, tehát tetszõleges -a határozott integrál definíciójában szereplõ feltételeket kielégítõ- integrálközelítõ összegeinek sorozata ugyanahhoz a számhoz konvergál. Az elõbbiek szerint ez a szám éppen $F(b)-F(a)$.
Ezzel a tétel bizonyítását befejeztük.

Példák

1. Példa

Számítsuk ki a következõ integrál értékét

$$\int_a^b x^2\,dx$$

Megoldás

Az x^2 függvény egyik primitív függvénye az $F(x)=\frac{x^3}{3}$ függvény. Ezért

```
> a:='a':b:='b':
> F:=x^3/3;
```

(3.1)

$$F := \frac{1}{3}x^3 \tag{3.1}$$

```
> Int(x^2,x=a..b)=subs(x=b,F)-subs(x=a,F);
```

$$\int_a^b x^2\,dx = \frac{1}{3}b^3 - \frac{1}{3}a^3 \tag{3.2}$$

A Maple-lel közvetlenül kiszámítva:

```
> Int(x^2,x=a..b)=int(x^2,x=a..b);
```

$$\int_a^b x^2\,dx = \frac{1}{3}b^3 - \frac{1}{3}a^3 \tag{3.3}$$

2. Példa

Számítsuk ki a következõ integrál értékét

$$\int_0^{\frac{2\pi}{3}} x\sin(x)\,dx$$

Megoldás

1. Megoldás

Elõször, parciális integrálással meghatározunk egy primitív függvényt, majd alkalmazzuk a Newton -Leibniz tételt.

```
> x:='x':
> with(student):
> Int(x*sin(x),x)=intparts(Int(x*sin(x),x),x);
```

$$\int x\sin(x)\,dx = -x\cos(x) - \int -\cos(x)\,dx \tag{3.4}$$

```
> lhs(%)=value(rhs(%));
```

$$\int x\sin(x)\,dx = -x\cos(x) + \sin(x) \tag{3.5}$$

Képezzük ezután az $F(b) - F(a)$ kifejezést. Helyettesítsünk a primitív függvényben az x helyére $\frac{2\pi}{3}$-at, majd 0-t, s képezzük az így nyert két érték különbségét!

```
> Int(x*sin(x),x = 0 .. 2*Pi/3)=subs(x=2*Pi/3,rhs(%))-subs(x=0,rhs
  (%));
```

$$\int_0^{\frac{2}{3}\pi} x\sin(x)\,dx = -\frac{2}{3}\pi\cos\left(\frac{2}{3}\pi\right) + \sin\left(\frac{2}{3}\pi\right) - \sin(0) \tag{3.6}$$

```
> lhs(%)=value(rhs(%));
```

(3.7)

$$\int_0^{\frac{2}{3}\pi} x\sin(x)\,dx = \frac{1}{3}\pi + \frac{1}{2}\sqrt{3} \qquad (3.7)$$

2. Megoldás

A parciális integrálás módszerét határozott integrál kiszámítására is alkalmazhatjuk:

```
> Int(x*sin(x),x=0..2*Pi/3)=intparts(Int(x*sin(x),x=0..2*Pi/3),x);
```

$$\int_0^{\frac{2}{3}\pi} x\sin(x)\,dx = \frac{1}{3}\pi - \int_0^{\frac{2}{3}\pi} -\cos(x)\,dx \qquad (3.8)$$

```
> lhs(%)=value(rhs(%));
```

$$\int_0^{\frac{2}{3}\pi} x\sin(x)\,dx = \frac{1}{3}\pi + \frac{1}{2}\sqrt{3} \qquad (3.9)$$

3. Megoldás

A Maple az integrál értékét közvetlenül is megadja:

```
> Int(x*sin(x),x=0..2*Pi/3)=int(x*sin(x),x=0..2*Pi/3);
```

$$\int_0^{\frac{2}{3}\pi} x\sin(x)\,dx = \frac{1}{3}\pi + \frac{1}{2}\sqrt{3} \qquad (3.10)$$

3. Példa

Számítsuk ki az alábbi integrált

$$\int_0^2 \sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx$$

Megoldás

Próbáljuk először a Maple segítségével megadni az integrál pontos értékét!

```
> Int(sqrt(ln(x+sqrt(1+x^2))/(1+x^2)),x=0..2);
```

$$\int_0^2 \sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx \qquad (3.11)$$

```
> value(%);
```

$$\int_0^2 \sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx \qquad (3.12)$$

A Maple közvetlenül nem tudja kiszámítani az integrál értékét. Próbáljunk segíteni! Nézzük a gyök alatti kifejezés számlálójának deriváltját!

```
> Diff(ln(x+(1+x^2)^(1/2)),x)=diff(ln(x+(1+x^2)^(1/2)),x);
```

$$\frac{d}{dx}\ln\left(x+\sqrt{1+x^2}\right)=\frac{1+\frac{x}{\sqrt{1+x^2}}}{x+\sqrt{1+x^2}} \tag{3.13}$$

Egyszerûsítsük a deriváltat!

```
> Diff(ln(x+(1+x^2)^(1/2)),x)=simplify(diff(ln(x+(1+x^2)^(1/2)),x)
  );
```

$$\frac{d}{dx}\ln\left(x+\sqrt{1+x^2}\right)=\frac{1}{\sqrt{1+x^2}} \tag{3.14}$$

A derivált éppen az integrandus nevezõje. Célszerû tehát a $t=\ln\left(x+\sqrt{1+x^2}\right)$ helyettesítés. A changevar eljárást a határozott integrál kiszámításakor is alkalmazhatjuk. Ekkor a helyettesítés során a határok is transzformálódnak, nem kell visszatérnünk az eredeti változóra. Elsõ lépésként elvégezzük a helyettesítést:

```
> Int(sqrt(ln(x+sqrt(1+x^2))/(1+x^2)),x=0..2)=changevar(t=ln(x+(1+
  x^2)^(1/2)),Int(sqrt(ln(x+sqrt(1+x^2))/(1+x^2)),x=0..2),t);
```

$$\int_0^2\sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx=\int_0^{\ln\left(2+\sqrt{5}\right)}\sqrt{t}\,dt \tag{3.15}$$

Számítsuk ki az integrál értékét:

```
> lhs(%)=value(rhs(%));
```

$$\int_0^2\sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx=\frac{2}{3}\ln\left(2+\sqrt{5}\right)^{(3/2)} \tag{3.16}$$

Az integrál értéke közelítõleg

```
> lhs(%)=evalf(rhs(%));
```

$$\int_0^2\sqrt{\frac{\ln\left(x+\sqrt{1+x^2}\right)}{1+x^2}}\,dx=1.156365321 \tag{3.17}$$

```
>
```